Endüstri Mühendisliğinde Bilgisayar Uygulamaları

GAMS

GAMS

- **GAMS**: General Algebraic Modeling System
- Doğrusal, Doğrusal Olmayan, Karma Tamsayılı optimizasyon problemleri modellenebilir.
- Büyük, karmaşık problemler modellenebilir.

GAMS

GAMS genellikle büyük boyutlu optimizasyon modellerinin çözümü için tasarlanmış bir yazılım olmasına karşın küçük modellere de açık halde çözüm üretme yeteneğine sahiptir.

Aşağıda verilen örneği inceleyelim;

Ürün 1 ve ürün 2 olmak üzere 2 farklı ürün imal eden bir firmanın aşağıdaki doğrusal programlama modelini ele aldığımızı varsayalım.

GAMS

$$Z_{maks} = 4x_1 + 5x_2$$

kısıtlar

$x_1 + x_2 \leq 60$ (I Numaralı kısıt)

$x_1 - 2x_2 \leq 40$ (II Numaralı kısıt)

$x_1 - x_2 \geq 0$ (III Numaralı kısıt)

$x_2 \leq 20$ (IV Numaralı kısıt)

$x_1, x_2 \geq 0$ (İşaret kısıtı)

GAMS

Bu modeli GAMS ile modelleyebilmek için önce değişkenlerimizi tanımlamalıyız.



GAMS

Tanımladığımız bu değişkenlerin pozitif olması gereklidir.



GAMS

Daha sonra amaç fonksiyonunu ve kısıtları tanımlarız.

GAMS kullanırken dikkat etmeniz gereken husus, kullanacağınız tüm denklem ve eşitsizlikleri önceden tanımlamanız gerektiğidir.



GAMS

Denklem ve eşitsizlikleri tanımladıktan sonra fonksiyonumuzu ve eşitsizliklerimizi yazmaya başlayabiliriz.

GAMS’de ilişkisel operatörler aşağıdaki şekilde tanımlanır.

= → =e=

≤ → =l=

≥ → =g=

Untitled_2.gms | Untitled_2.lst

```
variables
x1
x2
z;

positive variables
x1
x2;

equations
amac
kisit1
kisit2
kisit3
kisit4;

amac..    z =e= 4*x1+5*x2;
kisit1..  x1+x2 =l= 60;
kisit2..  x1-2*x2=l= 40;
kisit3..  x1-x2 =g= 0;
kisit4..  x2 =l= 20;
```

GAMS

Modelimizi GAMS’de tanımladık, fakat modelin çözümü için gerekli komutları vermemiz gereklidir.

Bunun için öncelikle modelimize bir isim vermeliyiz.

Bizim örneğimizin adı “ornek1” olursa aşağıdaki komutla modelimizin oluşturulması sağlanabilir.

```
variables
x1
x2
z;

positive variables
x1
x2;

equations
amac
kisit1
kisit2
kisit3
kisit4;

amac..    z =e= 4*x1+5*x2;
kisit1..  x1+x2 =l= 60;
kisit2..  x1-2*x2=l= 40;
kisit3..  x1-x2 =g= 0;
kisit4..  x2 =l= 20;

model ornek1/all/;
```

GAMS

Bu komut gereksiz gibi gözükse de aslında ileri seviyede, bir Gams modeli üstünden bir çok model çalıştırmak isteyen kullanıcılar için çok kullanışlıdır.

"/" işareti ile sınırlandırılan alana modele dahil edilmesi istenen denklem ve kısıtlar yazılarak model sınırlandırılabilir.

Bu özellik, aynı zamanda her fonksiyonun önce neden tanımlanması gerektiğini açıklamaktadır.

Biz modelimizde tüm kısıtları kullanmak istediğimiz için "all" (tümü) komutunu kullanacağız.

GAMS

Son olarak modeli çözmek için “solve” komutu çalıştırılır.

Bu komutun yapısı aşağıda verilen yapıya uygun olarak oluşturulmalıdır.

1. “Solve” komutu
2. Çözülmek istenen modelin adı
3. “using” komutu, bu komut kullanmak istediğimiz yöntemi seçim şansı sunar.

GAMS

4. Çözüm yöntemi; çözüm yönteminin model yapısına uygun seçilmesi gerekmektedir. GAMS içinde bulunan bazı yöntemler şunlardır.

   "lp" doğrusal programlama
   "nlp" doğrusal olmayan programlama
   "mip" tamsayılı programlama
   "rmip" gevşetilmiş tamsayılı programlama
   "minlp" tamsayılı doğrusal olmayan programlama
   "rminlp" gevşetilmiş tamsayılı doğrusal olmayan programlama
   "mpec" denge kısıtlı matematiksel modeller
   "cns" kısıtlanmış nonlinear sistemler

5. Amacınıza göre "minimizing" veya "maximizing" komutu

6. Optimize edilmek değişkenin adı (bu örnekte "z")

```
variables
x1
x2
z;

positive variables
x1
x2;

equations
amac
kisit1
kisit2
kisit3
kisit4;

amac..     z =e= 4*x1+5*x2;
kisit1.. x1+x2 =l= 60;
kisit2.. x1-2*x2=l= 40;
kisit3.. x1-x2 =g= 0;
kisit4.. x2 =l= 20;

model ornek1/all/;

solve ornek1 using lp maximizing z;
```

GAMS

Örnek: Kargo Yükleme Problemi

Bir kargo uçağı 5.000 kg'lık taşıma kapasitesine haizdir. Her uçuşta aşağıda sunulan kâr düzeyine göre farklı ağırlıklar söz konusudur.

	Ağırlık (kg)	**Toplam Kâr (Bin TL)**
1	1000	100
2	600	600
3	3000	400
4	2000	250

Her uçuş için kapasiteyi geçmeyecek şekilde toplam kârı maksimize eden GAMS kodunu yazınız.

GAMS

Kargo Yükleme Problemi-Çözüm

Eğer "i." kalem taşınacaksa xi = 1

Eğer "i." kalem taşınmayacaksa xi = 0

Maks Z = 100X1 + 600X2 + 400X3 + 250X4

kısıtlar

1000X1 + 600X2 + 3000X3 + 2000X4 ≤ 5000 (kapasite kısıtı)

X1, X2, X3, X4 = (0, 1) (0-1 Tam sayı kısıtı)

Değişkenler Tamsayı istenseydi???

GAMS

```
Variables
x1, x2, x3, x4, z;
Binary Variables
x1, x2, x3, x4;
Equations
Amac
Kisit;
Amac..     z =E= 100*x1+600*x2+400*x3+250*x4;
Kisit..    1000*x1+600*x2+3000*x3+2000*x4 =L= 5000;
Model Ornek2 /all/;
Solve Ornek2 Maximizing z  Using mip;
```

GAMS

Örnek: Aşağıda verilen 0-1 tamsayılı modelin çözümünü **GAMS** ile bulunuz.

Min z = x1 + x2 + x3 + x4 + x5 + x6 + x7 + x8

kısıtlar

x1 + x2 + x7 ≥ 1 (Yerleşke 1)

x1 + x2 + x6 + x8 ≥ 1 (Yerleşke 2)

x3 + x4 + x8 ≥ 1 (Yerleşke 3)

x3 + x4 + x5 + x7 ≥ 1 (Yerleşke 4)

x4 + x5 + x6 + x8 ≥ 1 (Yerleşke 5)

x2 + x5 + x6 + x7 ≥ 1 (Yerleşke 6)

x1 + x4 + x6 + x7 + x8 ≥ 1 (Yerleşke 7)

x2 + x3 + x5 + x7 + x8 ≥ 1 (Yerleşke 8)

xj = 0 veya 1 (j = 1, 2, 3, 4, 5, 6, 7, 8)

```
Variables
x1, x2, x3, x4, x5, x6, x7, x8, z;
Binary Variables
x1, x2, x3, x4, x5, x6, x7, x8;
Equations
Amac
Kisit1, Kisit2, Kisit3, Kisit4, Kisit5, Kisit6, Kisit7, Kisit8;
Amac..     z =E= x1+x2+x3+x4+x5+x6+x7+x8;
Kisit1..    x1+x2+x7 =G= 1;
Kisit2..    x1+x2+x6+x8 =G= 1;
Kisit3..    x3+x4+x8 =G= 1;
Kisit4..    x3+x4+x5+x7 =G= 1;
Kisit5..    x4+x5+x6+x8 =G= 1;
Kisit6..    x2+x5+x6+x7 =G= 1;
Kisit7..    x1+x4+x6+x7+x8 =G= 1;
Kisit8..    x2+x3+x5+x7+x8 =G= 1;
Model Ornek3 /all/;
Solve Ornek3 Minimizing z  Using mip;
```

ÖRNEK: Aşağıda verilen non-lineer matematiksel modelin çözümünü **GAMS** ile bulunuz.

$$\text{Max } Z = x_1 + 2x_3 + x_2x_3 - (x_1^2 + x_2^2 + x_3^2)$$

kısıtlar

$$x_1^2 + x_2^2 \geq 1$$

$$x_3 \leq 1$$

$$x_1, x_2, x_3 \geq 0$$

! Değişkenler tamsayı istense idi???

ÇÖZÜM

```
variables x1, x2, x3, z;
equations F, k1, k2;
F.. z=e= x1+2*x3+x2*x3 - x1*x1 - x2*x2 - x3*x3;
k1.. x1*x1 + x2*x2 =g= 1;
k2.. x3 =l= 1;
model nonlin /all/;
solve nonlin using nlp maximizing z;
```

GAMS

- Modeli bu şekilde yazmak doğru olmasına rağmen, bu yaklaşım kullanıldığında büyük kapsamlı modellerin modellenmesinde problemler çıkabilir. Bu sebeple yazım üstünde bazı değişiklikler yapılabilir.

- İndeks Tanımlanması: Modelde verilen iki ayrı değişkenin ayrı ayrı tanımlanmasına gerek yoktur. Bunun yerine bir x değişkeni tanımlanıp, iki ayrı indeks atanabilir.

GAMS

GAMS KAPALI MODEL YAPISI

GAMS model terminolojisinde matematiksel model bileşenlerine karşılık gelen kavramlar aşağıdaki tabloda verilmiştir.

Matematiksel model bileşeni	GAMS modeli karşılığı
• İndisler	• Sets
• Parametreler	• Parameters
○ Amaç fonksiyonu katsayıları	
○ Kısıt katsayıları (teknolojik katsayılar)	
○ Sağ taraf değerleri	
• Karar değişkenleri	• Variables
• Eşitlikler ve Eşitsizlikler	• Equations
○ Amaç fonksiyonu	
○ Kısıtlar	

GAMS

Örnek: Bir işletme dört farklı masa üretmektedir. Her bir masa 2 farklı işlemden geçmektedir. Bunlar; kesim ve montajdır. Aşağıdaki tabloda her bir masanın işlem süreleri ve satış fiyatları verilmiştir. İşletmenin kârını maksimum yapacak ürün karışımını **GAMS** ile bulunuz.

	M1	M2	M3	M4	Kapasite
Kesim	4	9	7	10	6000
Montaj	1	1	3	40	4000
Fiyat	12	20	18	40	

```
Sets  masa / m1, m2, m3, m4 /
      islem / kesim, montaj /
Parameters
  kapasite(islem)/ kesim = 6000, montaj = 4000 /
  fiyat(masa)/ m1 = 12, m2 = 20, m3 = 18, m4 = 40 /
Table sure(islem,masa)
              m1    m2    m3    m4
 kesim         4     9     7    10
 montaj        1     1     3    40
Variables
  miktar(masa)
  kar
Integer Variable  miktar
Equations
  kap(islem)
  TopKar;
kap(islem)..   sum(masa, sure(islem,masa)*miktar(masa)) =l= kapasite(islem);
TopKar..       kar =e= sum(masa, fiyat(masa)*miktar(masa));
Model ukp / all /; Solve ukp maximizing kar using mip;
```

GAMS

Örnek: 2 farklı fabrika ve 3 farklı müşterinin bulunduğu bir ulaştırma problemine ait mesafeler, fabrika kapasiteleri ve müşteri talepleri aşağıdaki tabloda verilmiştir. Taşıma firması her 1000km için 90PB talep ettiğine göre minimum maliyetli taşıma modelini **GAMS** kapalı form ile bulunuz.

Fabrikalar	**Müşteriler**			**Kapasite**
	İzmir	**Konya**	**Gaziantep**	
Ankara	579	258	673	350
İstanbul	561	668	1126	600
Talep	325	300	275	

```
Sets
i   fabrikalar   / Ankara, Istanbul /
j   musteriler   / Izmir, Konya, Gaziantep / ;
Parameters
a(i)  i fabrikasının kapasitesi
        /    Ankara     350
             Istanbul   600  /
b(j)  j musterisinin talebi
        /    Izmir        325
             Konya        300
             Gaziantep  275  / ;
Table d(i,j)  kilometre cinsinden mesafeler
                Izmir     Konya     Gaziantep
Ankara          579       258          673
Istanbul        561       668          1126 ;
```

```
Scalar f  1000 km icin birim ulastirma maliyeti  /90/ ;
Parameter c(i,j)  1000 km icin i fabrikasından j musterisine
                  birim ulastirma maliyetleri ;
                 c(i,j) = f * d(i,j) / 1000 ;
Variables
x(i,j)  i fabrikasindan j musterisine tasima miktari
z       toplam ulastirma maliyeti ;
Positive Variable x ;
Equations
amac        amac fonksiyonu
arz(i)      i fabrikasinin kapasite kisiti
talep(j)    j musterisinin talep kisiti ;
amac ..     z  =e=  sum((i,j), c(i,j)*x(i,j)) ;
arz(i) ..   sum(j, x(i,j))  =l=  a(i) ;
talep(j) .. sum(i, x(i,j))  =g=  b(j) ;
Model ulastirma /all/ ;    Solve ulastirma using lp minimizing z ;
Display x.l, x.m ;
```

Tesis Yerleşim Problemi

Bir firma 4 bölge bayisinin talebini karşılayabilmek için yeni fabrika(lar) kurmak istemektedir. Bu amaçla 3 adet alternatif yer belirlemiştir. Firma her bir alternatif yer için sabit kuruluş maliyetlerini sırasıyla 91, 70 ve 24 Pb ve üretim kapasitelerini ise haftalık olarak 39, 35, 31 ton olarak belirlemiştir. Bayilerin haftalık talepleri ise 15, 17, 22 ve 12 ton olarak belirlenmiştir. Alternatif tesis yerleri ile bayiler arasındaki bir ton ürün taşıma maliyetleri aşağıdaki tabloda verilmiştir. Buna göre toplam maliyeti minimize edecek kuruluş yer(ler)ini **GAMS** ile belirleyiniz.

	Bayi 1	Bayi 2	Bayi 3	Bayi 4
Tesis 1	6	2	6	7
Tesis 2	4	9	5	3
Tesis 3	8	8	1	5

Tesis Yerleşim Problemi

Karar Değişkenleri : ???

X_{ij} = i tesisinden j bayisine gönderilecek miktar (i=1,2,3; j=1,2,3,4) $X_{ij} \geq 0$

Y_i = i tesisinin kurulup kurulmama kararı (i=1,2,3) Y_i = 0 veya 1

Tesis Yerleşim Problemi

Amaç Fonksiyonu: ???

$$
\begin{aligned}
Z_{min} = {} & 6X_{11} + 2X_{12} + 6X_{13} + 7X_{14} + \\
& 4X_{21} + 9X_{22} + 5X_{23} + 3X_{24} + \\
& 8X_{31} + 8X_{32} + 1X_{33} + 5X_{34} + \\
& 91Y_1 + 70Y_2 + 24Y_3
\end{aligned}
$$

Tesis Yerleşim Problemi

Kısıtlar: ???

$X11 + X21 + X31 >= 15$

$X12 + X22 + X32 >= 17$

$X13 + X23 + X33 >= 22$

$X14 + X24 + X34 >= 12$

$X11 + X12 + X13 + X14 <= 39Y1$

$X21 + X22 + X23 + X24 <= 35Y2$

$X31 + X32 + X33 + X34 <= 31Y3$

$Xij >= 0$ $Yi = 0$ veya 1

```
Sets
t Tesis /1,2,3/
b Bayi /1,2,3,4/
Parameter Sa_mal(t) /1 91, 2 70, 3 24/;
Parameter Kap(t) /1 39, 2 35, 3 31/;
Parameter Talep(b) /1 15, 2 17, 3 22, 4 12/;
Table c(t,b)
   1   2   3   4
1  6   2   6   7
2  4   9   5   3
3  8   8   1   5 ;
```

```
Positive variable x(t,b)
Binary variable y(t)
Variable z;
Equations
k1, k2(b), k3(t);
k1.. z =e= sum((t,b),c(t,b)*x(t,b))+ sum(t, Sa_mal(t)*y(t));
k2(b).. sum(t,x(t,b)) =g= Talep(b);
k3(t).. sum (b,x(t,b)) =l= Kap(t)*y(t);
Model TesYer /all/;
Solve TesYer minimizing z using mip;
```

Dağıtım Modeli

Bir otomobil firmasının 2 adet fabrikası ve bu fabrikalarda ürettiği 2 otomobil modeli(A, B) bulunmaktadır. Üretilen tüm otomobiller 3 adet bayiye gönderilmektedir. 1. fabrika bir ayda A modelinden 50 ad. B modelinden 70 ad. üretebilmektedir. 2. fabrika bir ayda A modelinden 60 ad. B modelinden 40 ad. üretebilmektedir. 1. bayi aylık olarak A ve B modellerinden sırasıyla 40 ve 30 ad., 2. bayi A ve B modellerinden 30 ve 30 ad., 3. bayi ise A ve B modellerinden 40 ve 50 ad. otomobil talep etmektedir. Fabrikalar ve bayiler arasındaki bir otomobil taşıma maliyeti(pb) aşağıdaki tabloda verilmiştir. Toplam taşıma maliyetini minimum yapan dağıtım planını bulunuz.

	Bayi1	Bayi2	Bayi3
Fabrika1	8	7	10
Fabrika2	9	11	4

X_{ijk} = i fabrikasından j bayisine k modelden gönderilen miktar

$Z_{min} = 8X_{111} + 8X_{112} + 7X_{121} + 7X_{122} + 10X_{131} + 10X_{132} + 9X_{211} + 9X_{212} + 11X_{221} + 11X_{222} + 4X_{231} + 4X_{232}$

kısıtlar

$X_{111} + X_{121} + X_{131} = 50$

$X_{112} + X_{122} + X_{132} = 70$

$X_{211} + X_{221} + X_{231} = 60$

$X_{212} + X_{222} + X_{232} = 40$

$X_{111} + X_{211} = 40$

$X_{112} + X_{212} = 30$

$X_{121} + X_{221} = 30$

$X_{122} + X_{222} = 30$

$X_{131} + X_{231} = 40$

$X_{132} + X_{232} = 50$ Tüm $X_{ijk} \geq 0$

Sets

i fabrikalar /1, 2/

j bayiler /1, 2, 3/

k modeller /1, 2/;

Parameters

a(i,k) i fabrikasinin k model kapasitesi

/1.1 50, 1.2 70, 2.1 60, 2.2 40/

b(j,k) j bayisinin k model talebi

/1.1 40, 1.2 30, 2.1 30, 2.2 30, 3.1 40, 3.2 50/

c(i,j,k) i fabrikasindan j bayisine k modelden tasima maliyeti

/1.1.1 8, 1.1.2 8, 1.2.1 7, 1.2.2 7, 1.3.1 10

1.3.2 10, 2.1.1 9, 2.1.2 9, 2.2.1 11, 2.2.2 11

2.3.1 4, 2.3.2 4/;

```
Variables
x(i,j,k) i fabrikasindan j bayisine k modelden tasinacak miktar
z        toplam tasima maliyeti;
Positive variable x;
Equations
amac        amac fonksiyonu
arz(i,k)    i fabrikasinin k model kapasite kisiti
talep(j,k)  j bayisinin k model talep kisiti;
amac..      z =e= sum((i,j,k),c(i,j,k)*x(i,j,k));
arz(i,k)..  sum(j,x(i,j,k)) =e= a(i,k);
talep(j,k).. sum(i,x(i,j,k)) =e= b(j,k);
Model dagitim /all/;
Solve dagitim using lp minimizing z;
Display x.l, x.m ;
```

ÖRNEK: ABC işletmesi 4 peryotluk üretim çizelgesi hazırlamak istemektedir. İşletme her peryot için ürünlerini hem normal mesai hem de fazla mesaide üretebilmektedir. Ayrıca her peryot için maksimum 70 adet ürün için stok alanı mevcuttur. Aşağıdaki tabloda her peryot için talep, normal mesai maliyet ve kapasite, fazla mesai maliyet ve kapasite değerleri verilmiştir. Başlangıç stoğu 15 adet ve stokta tutma maliyeti her birim için 1,5 pb olduğuna göre her peryodun talebini karşılayacak minimum maliyetli üretim çizelgesini GAMS yardımı ile bulunuz.

Peryot	Talep	Normal Mesai Kapasitesi	Normal Mesai Maliyeti	Fazla Mesai Kapasitesi	Fazla Mesai Maliyeti
1	130	100	6	60	8
2	80	100	4	65	6
3	125	100	8	70	10
4	195	100	9	60	11

Karar Değişkenleri

x_t = t. peryotta normal mesaide üretilen ürün miktarı
(t = 1,2,3,4), $x_t \geq 0$ ve tamsayı

y_t = t. peryotta fazla mesaide üretilen ürün miktarı
(t = 1,2,3,4), $y_t \geq 0$ ve tamsayı

I_t = t. peryot sonunda elde kalan stok miktarı
(t = 0,1,2,3,4), $I_t \geq 0$ ve tamsayı

Amaç Fonksiyonu

$$Z_{min} = (6x_1 + 4x_2 + 8x_3 + 9x_4) +$$
$$(8y_1 + 6y_2 + 10y_3 + 11y_4) +$$
$$(1.5I_0 + 1.5I_1 + 1.5I_2 + 1.5I_3 + 1.5I_4)$$

Kısıtlar

Her peryot için üretim miktarı kapasiteyi aşamaz.

Normal Mesai	Fazla Mesai
$x_1 \leq 100$	$y_1 \leq 60$
$x_2 \leq 100$	$y_2 \leq 65$
$x_3 \leq 100$	$y_3 \leq 70$
$x_4 \leq 100$	$y_4 \leq 60$

Her peryot için stokta tutulacak ürün miktarı 70’i geçemez.

$I_1 \leq 70$

$I_2 \leq 70$

$I_3 \leq 70$

$I_4 \leq 70$

Her peryot sonundaki stok miktarı =

Başlangıç stoğu + Üretim – Talep

$I_0 = 15$

$I_1 = I_0 + (x_1 + y_1) - 130$

$I_2 = I_1 + (x_2 + y_2) - 80$

$I_3 = I_2 + (x_3 + y_3) - 125$

$I_4 = I_3 + (x_4 + y_4) - 195$

Her peryodun talebi mutlaka karşılanmak zorunda;

$I_0 + (x_1 + y_1) \geq 130$

$I_1 + (x_2 + y_2) \geq 80$

$I_2 + (x_3 + y_3) \geq 125$

$I_3 + (x_4 + y_4) \geq 195$

$$Z_{min} = 6x_1 + 4x_2 + 8x_3 + 9x_4 + 8y_1 + 6y_2 + 10y_3 + 11y_4 + 1.5I_0 + 1.5I_1 + 1.5I_2 + 1.5I_3 + 1.5I_4$$

kısıtlar

$x_1 \leq 100$	$y_1 \leq 60$	$I_1 \leq 70$
$x_2 \leq 100$	$y_2 \leq 65$	$I_2 \leq 70$
$x_3 \leq 100$	$y_3 \leq 70$	$I_3 \leq 70$
$x_4 \leq 100$	$y_4 \leq 60$	$I_4 \leq 70$

$I_0 = 15$

$I_1 = I_0 + (x_1 + y_1) – 130$

$I_2 = I_1 + (x_2 + y_2) – 80$

$I_3 = I_2 + (x_3 + y_3) – 125$

$I_4 = I_3 + (x_4 + y_4) – 195$

$I_0 + (x_1 + y_1) \geq 130$

$I_1 + (x_2 + y_2) \geq 80$

$I_2 + (x_3 + y_3) \geq 125$

$I_3 + (x_4 + y_4) \geq 195$ $x_t, y_t, I_t \geq 0$ ve tamsayı $t = 0, 1, 2, 3, 4, 5$

GAMS-ÇÖZÜM

Sets t Periyot /0*4/;

Parameters

Talep(t) /1 130, 2 80, 3 125, 4 195/

Knorm(t) /1 100, 2 100, 3 100, 4 100/

Mnorm(t) /1 6, 2 4, 3 8, 4 9/

Kfazla(t) /1 60, 2 65, 3 70, 4 60/

Mfazla(t) /1 8, 2 6, 3 10, 4 11/;

Integer Variable x(t) t. peryotta normal mesaide üretilen ürün miktarı

Integer Variable y(t) t. peryotta fazla mesaide üretilen ürün miktarı

Integer Variable I(t) t. peryot sonunda elde kalan stok miktarı

Variable Z   Amaç Fonksiyonu;

Equations

Kisit_1, Kisit_2(t), Kisit_3(t), Kisit_4(t), Kisit_5(t), Kisit_6(t), Kisit_7(t);

GAMS-ÇÖZÜM

```
Kisit_1..
sum(t$(ord(t)>1),Mnorm(t)*x(t))+sum(t$(ord(t)>1),Mfazla(t)*y(t))+
sum(t$(ord(t)>0),1.5*I(t)) =E= Z;
Kisit_2(t)$(ord(t)>1).. x(t) =L= Knorm(t);
Kisit_3(t)$(ord(t)>1).. y(t) =L= Kfazla(t);
Kisit_4(t)$(ord(t)=1).. I(t) =E= 15;
Kisit_5(t)$(ord(t)>1).. I(t) =L= 70;
Kisit_6(t)$(ord(t)>1).. I(t) =E= I(t-1)+(x(t)+y(t))-Talep(t);
Kisit_7(t)$(ord(t)<=4).. I(t)+x(t+1)+y(t+1) =G= Talep(t+1);
Model cizelge /all/;
Solve cizelge using mip minimizing Z;
```

$$Z_{min} = 10x_{12} + 8x_{13} + 9x_{14} + 7x_{15} + 10x_{21} + 10x_{23} + 5x_{24} + 6x_{25} + 8x_{31} + 10x_{32} + 8x_{34} + 9x_{35} + 9x_{41} + 5x_{42} + 8x_{43} + 6x_{45} + 7x_{51} + 6x_{52} + 9x_{53} + 6x_{54}$$

kısıtlar

$$\begin{aligned}
x_{12} + x_{13} + x_{14} + x_{15} &= 1 \\
x_{21} + x_{23} + x_{24} + x_{25} &= 1 \\
x_{31} + x_{32} + x_{34} + x_{35} &= 1 \\
x_{41} + x_{42} + x_{43} + x_{45} &= 1 \\
x_{51} + x_{52} + x_{53} + x_{54} &= 1
\end{aligned}$$

$$\begin{aligned}
x_{21} + x_{31} + x_{41} + x_{51} &= 1 \\
x_{12} + x_{32} + x_{42} + x_{52} &= 1 \\
x_{13} + x_{23} + x_{43} + x_{53} &= 1 \\
x_{14} + x_{24} + x_{34} + x_{54} &= 1 \\
x_{15} + x_{25} + x_{35} + x_{45} &= 1
\end{aligned}$$

$x_{ij} = 0$ veya 1

$u_i \geq 0$

$$\begin{aligned}
u_2 - u_3 + 5x_{23} &\leq 4 \\
u_2 - u_4 + 5x_{24} &\leq 4 \\
u_2 - u_5 + 5x_{25} &\leq 4 \\
u_3 - u_2 + 5x_{32} &\leq 4 \\
u_3 - u_4 + 5x_{34} &\leq 4 \\
u_3 - u_5 + 5x_{35} &\leq 4 \\
u_4 - u_2 + 5x_{42} &\leq 4 \\
u_4 - u_3 + 5x_{43} &\leq 4 \\
u_4 - u_5 + 5x_{45} &\leq 4 \\
u_5 - u_2 + 5x_{52} &\leq 4 \\
u_5 - u_3 + 5x_{53} &\leq 4 \\
u_5 - u_4 + 5x_{54} &\leq 4
\end{aligned}$$

```
SETS
I Nodes          /1*5/
ALIAS(I,J);
TABLE Cost(I,J) ****
VARIABLE Z;
BINARY VARIABLE X(I,J);
POSITIVE VARIABLE U(I);
X.fx(I,I) = 0;
EQUATIONS
OBJ, CONST1, CONST2, CONST3;
OBJ..              Z =E= SUM((I,J), Cost(I,J)*X(I,J));
CONST1(I)..        SUM(J,X(I,J)) =E= 1;
CONST2(J)..        SUM(I,X(I,J)) =E= 1;
CONST3(I,J) $ ((ORD(I)<>ORD(J)) AND (ORD(I)<>1))..
U(I) - U(J) + 5*X(I,J) =L= 4;
MODEL ORNEK10 /all/;
SOLVE ORNEK10 USING MIP MINIMIZING Z;
```

Aşağıda verilen karma tamsayılı matematiksel modeli **GAMS** ile çözünüz.

$$\sum_{i=1}^{5}\sum_{j=1}^{5} A_{ij} * (XP_{ij} + XN_{ij})$$

kısıtlar

$$X_i - X_j + M * Z_{ij} \geq 0.5 * \left(L_i + H_j + D_{ij}\right) \qquad \forall i, j \; ve \; i \neq j$$

$$X_j - X_i + M * \left(1 - Z_{ij}\right) \geq 0.5 * \left(L_i + H_j + D_{ij}\right) \qquad \forall i, j \; ve \; i \neq j$$

$$XP_{ij} - XN_{ij} = X_i - X_j \qquad \forall i, j \; ve \; i \neq j$$

$$X_i, X_j, XP_{ij}, XN_{ij} \geq 0$$

$$Z_{ij} = 0 \; veya \; 1$$

M = Büyük bir sayı(=1000)

PARAMETRELER

A_{ij}					
	1	2	3	4	5
1	0	8	4	15	0
2	4	0	3	2	2
3	4	1	0	5	0
4	5	4	5	0	1
5	0	0	0	2	0

D_{ij}					
	1	2	3	4	5
1	0	10	20	15	10
2	10	0	10	5	10
3	20	10	0	5	5
4	15	5	5	0	10
5	10	10	5	10	0

PARAMETRE L_i

1 20
2 10
3 10
4 20
5 15

PARAMETRE H_j

1 10
2 10
3 10
4 10
5 10

ÇÖZÜM

```
SETS
I Nodes  /1*5/
ALIAS (I,J);
TABLE A(I,J)***
TABLE D(I,J)***
PARAMETER L(I)***
PARAMETER H(J)***
PARAMETER BigM /1000/;
VARIABLES OBJFUNC;
POSITIVE VARIABLES X(I), XP(I,J), XN(I,J);
BINARY VARIABLES Z(I,J);
EQUATIONS
OBJ, C1, C2, C3;
OBJ.. OBJFUNC =E= SUM( (I,J), A(I,J)*(XP(I,J)+XN(I,J)) );
C1(I,J)$(ORD(I)<>ORD(J))..  X(I)-X(J)+BigM*Z(I,J) =G= 0.5*(L(I)+H(J)+D(I,J));
C2(I,J)$(ORD(I)<>ORD(J))..  X(J)-X(I)+BigM*(1-Z(I,J)) =G= 0.5*(L(I)+H(J)+D(I,J));
C3(I,J)$(ORD(I)<>ORD(J))..  XP(I,J)-XN(I,J) =E= X(I)-X(J);
MODEL ORNEK11 /all/;
SOLVE ORNEK11 USING MIP MINIMIZING OBJFUNC;
```

Aşağıda kapalı formda verilen **GAMS** kodunda bazı yazım yanlışları vardır. Modelin düzeltilmiş halini yazınız ve çözümünü bulunuz.

```
SETS t / 1, 2, 3, 4, 5, 6, 7, 8 /
PARAMETER x1(i) /1 2, 2 3, 3 3, 4 3, 5 5, 6 5, 7 6, 8 7/
PARAMETER x2(i) /1 30, 2 60, 3 70, 4 60, 5 80, 6 90, 7 100, 8 100/
PARAMETER x3(i) /1 1, 2 6, 3 7, 4 3, 5 5, 6 9, 7 8, 8 17/
PARAMETER y_hat(i) /1 10, 2 20, 3 30, 4 20, 5 40, 6 50, 7 60, 8 70/
VARIABLES
a, b, c, y, e, obj
EQUATION mod(t), residual(t), objective
mod a*x1(t)*x1(t)-b/x3(t)-c/x2(t)+(-y(t)*y(t)) = y(t)
residual e(t) = y(t)-yhat(t)
objective(t) obj = sum(t,e(t)*e(t));
MODEL QUIZ;
SOLVE QUIZ USING MIP MINIMIZING obj;
```

Aşağıda kapalı formda verilen **GAMS** kodunun açık halini yazınız.

```
Set t / 1*5 /
Scalars
rho / 8 /, alpha / 6 /, wage / 100 /
Parameters si(t) / 1 10 /, wi(t) / 1 20 /, sf(t) / 5 100 /, d(t) / 1 100,
2 200, 3 300,  4 400, 5 200 /
Positive Variables p(t), s(t), u(t), w(t), h(t), f(t)
Variables phi
Equations
cb(t), wb(t), wd(t), obj;
cb(t).. s(t) =e= s(t-1) + p(t) - d(t) - u(t-1) + u(t) + si(t);
wb(t).. w(t) =e= w(t-1) - f(t) + h(t) + wi(t);
wd(t).. w(t) =g= p(t)/rho + (1 + 1/alpha)*h(t);
obj.. phi =e= sum(t, 10*s(t) + 30*u(t) + (wage + sf(t))*w(t) );
Model jobs/ all /; Solve jobs minimizing phi using lp;
```

GAMS – Çözüm Parametreleri

```
isim.optcr=0;
isim.reslim=3600;
isim.iterlim=2e9;
isim.limrow=100000;
isim.limcol=100000;
```

Aşağıda verilen matematiksel modelin açık halini yazınız ve GAMS ile çözerek optimum sonucu bulunuz.

$$Z_{max} = \sum_{i=1}^{3} 400.D_i - 280O_1B_1 - 390O_2B_1 - 110C_1B_1 - 180C_2B_1 - 130C_3B_1 - 290O_1B_2 - 400O_2B_2 - 90C_1B_2 - 190C_2B_2 - 130C_3B_2 - 310O_1B_3 - 430O_2B_3 - 100C_1B_3 - 200C_2B_3 - 130C_3B_3 - 10(O_1S_1 + O_2S_1 + C_1S_1 + C_1S_1 + C_2S_1 + C_3S_1 + O_1S_2 + O_2S_2 + C_1S_2 + C_2S_2 + C_3S_2 + O_1S_3 + O_2S_3 + C_1S_3 + C_2S_3 + C_3S_3)$$

kısıtlar

(1) $O_jS_i = O_jS_{i-1} + O_jB_i - O_jU_i \qquad j = 1,2 \qquad i = 1,2,3$

(2) $C_jS_i = C_jS_{i-1} + C_jB_i - C_jU_i \qquad j = 1,2,3 \qquad i = 1,2,3$

(3) $O_jS_0 = 0 \qquad j = 1,2$

(4) $C_jS_0 = 0 \qquad j = 1,2,3$

(5) $O_jS_i \leq 300 \qquad i = 1,2,3 \qquad j = 1,2$

(6) $C_jS_i \leq 300 \qquad i = 1,2,3 \qquad j = 1,2,3$

(7) $O_1U_i + O_2U_i \leq 190 \qquad i = 1,2,3$

(8) $C_1U_i + C_2U_i + C_3U_i \leq 270 \qquad i = 1,2,3$

(9) $3 \leq \frac{3.1O_1U_i + 2.4O_2U_i + 7.2C_1U_i + 5.8C_2U_i + 6.1C_3U_i}{D_i} \leq 6 \qquad i = 1,2,3$

(10) $O_1U_i + O_2U_i + C_1U_i + C_2U_i + C_3U_i - D_i = 0 \qquad i = 1,2,3$

$O_jU_i \geq 0 \qquad i = 1,2,3 \qquad j = 1,2$

$C_jU_i \geq 0 \qquad i = 1,2,3 \qquad j = 1,2,3$

$D_i \geq 0 \qquad i = 1,2,3$